

TimeLED, **PCF8583SetDateLED**, **PCF8583SetAlarmLED** i **PCF8583ClearAlarm**. Rozpoczniemy od procedury **PCF8583SetTimeLED** ustawiającej czas (Przykład 10.72).

Przykład 10.72. Procedura ustawiająca czas (wyświetlacz LED i klawiatura sterowane sekwencyjnie)

```

PCF8583SetTimeLED:          ;USTAWIENIE CZASU
; [r15, r16, r17]
lds   r16,Flagi3           ;do r16 zawartość rejestru flag własnych Flagi3
sbrs  r16,SettingTime_Flg3 ;jeśli SettingTime_Flg3=1, to pomiń wyjście z procedury
ret

call  Load0ToDispMem       ;załadowanie kodów zera do pamięci wyświetlacza LED

call  PCF8583SettingHours  ;ustawianie godzin
ldi   r16,BlinkingMinutes_const ;do r16 liczba wskazująca, które moduły LED mają pulsować
sts   BlinkingLEDModule,r16  ;zapamiętanie nowej wartości
call  Hold2                 ;opóźnienie wywołania procedury ustawiającej minuty
.DB   1,0
call  PCF8583SettingMinutes ;ustawianie minut
clr   r16                   ;zerowanie r16
sts   BlinkingLEDModule,r16 ;zapamiętanie nowej wartości - kasowanie pulsowania

call  PCF8583Stop          ;zatrzymanie zegara
call  BlockINT3            ;zablokowanie przerwania INT3

lds   r16,TimeHours        ;do r16 ustawiona liczba godzin
call  ConvertToBCD         ;konwersja do formatu BCD
sts   TimeHours,r16        ;zapamiętanie liczby godzin w formacie BCD
ldi   XL,PCF8583Hours      ;do r16 adres rejestru godzin w układzie PCF8583
call  PCF8583Write         ;zapisanie liczby godzin czasu do układu PCF8583

lds   r16,TimeMinutes      ;do r16 ustawiona liczba minut
call  ConvertToBCD         ;konwersja do formatu BCD
sts   TimeMinutes,r16     ;zapamiętanie liczby minut w formacie BCD
ldi   XL,PCF8583Minutes    ;do r16 adres rejestru minut w układzie PCF8583
call  PCF8583Write         ;zapisanie liczby minut czasu do układu PCF8583

ldi   XL,PCF8583Seconds    ;do r16 adres rejestru sekund w układzie PCF8583
ldi   r16,0                ;do r16 liczba sekund ustawianego czasu
call  PCF8583Write         ;zapisanie liczby sekund czasu do układu PCF8583
ldi   XL,01                ;do r16 adres rejestru PCF8583 dziesiętnych i setnych sekundy
ldi   r16,0                ;do r16 liczba sekund ustawianego czasu
call  PCF8583Write         ;zapisanie liczby sekund czasu do układu PCF8583

call  UnblockINT3         ;odblokowanie przerwania INT3
call  PCF8583Start        ;uruchomienie zegara

ldi   r16,StartValueTCNT2 ;do r16 standardowy czas odświeżania wyświetlacza LED
sts   LEDRefreshTime,r16  ;zapamiętanie nowej wartości
call  ClearSettingTimeFlag_Flg3 ;zerowanie flagi informującej o wejściu w tryb ustawiania czasu
call  SetDispTimeFlag_Flg3 ;ustawienie flagi informującej o wyświetleniu czasu

ret

```

Procedura **PCF8583SetTimeLED** rozpoczyna działanie od sprawdzenia stanu flagi własnej **SettingTime_Flg3**. Jeśli jest ustawiona, rozkaz powrotu jest pomijany. Flaga **SettingTime_Flg3** ustawiana jest w procedurze **CheckIfSettingMode** (Przykład 10.68), wywoływanej cyklicznie przez program obsługi przerwania **ExtInt3**, podobnie jak pozostałe dwie flagi: **SettingDate_Flg3** oraz **SettingAlarm_Flg3**, testo-